Lesson No. 15

Parameter Passing Through Stack

Due to the limited number of registers, parameter passing by registers is constrained in two ways. The maximum parameters a subroutine can receive are seven when all the general registers are used. Also, with the subroutines are themselves limited in their use of registers, and this limited increases when the subroutine has to make a nested call thereby using certain registers as its parameters. Due to this, parameter passing by registers is not expandable and generalizable. However this is the fastest mechanism available for passing parameters and is used where speed is important.

Considering stack as an alternate, we observe that whatever data is placed there, it stays there, and across function calls as well. For example the bubble sort subroutine needs an array address and the count of elements. If we place both of these on the stack, and call the subroutine afterwards, it will stay there. The subroutine is invoked with its return address on top of the stack and its parameters beneath it.

To access the arguments from the stack, the immediate idea that strikes is to pop them off the stack. And this is the only possibility using the given set of information. However the first thing popped off the stack would be the return address and not the arguments. This is because the arguments were first pushed on the stack and the subroutine was called afterwards. The arguments cannot be popped without first popping the return address. If a heaving thing falls on someone’s leg, the heavy thing is removed first and the leg is not pulled out to reduce the damage. Same is the case with our parameters on which the return address has fallen.

To handle this using PUSH and POP, we must first pop the return address in a register, then pop the operands, and push the return address back on the stack so that RET will function normally. However so much effort doesn’t seem to pay back the price. Processor designers should have provided a logical and neat way to perform this operation. They did provided a way and infact we will do this without introducing any new instruction.

Recall that the default segment association of the BP register is the stack segment and the reason for this association had been deferred for now. The reason is to peek inside the stack using the BP register and read the parameters without removing them and without touching the stack pointer. The stack pointer could not be used for this purpose, as it cannot be used in an effective address. It is automatically used as a pointer and cannot be explicitly used. Also the stack pointer is a dynamic pointer and sometimes changes without telling us in the background. It is just that whenever we touch it, it is where we expect it to be. The base pointer is provided as a replacement of the stack pointer so that we can peek inside the stack without modifying the structure of the stack.

When the bubble sort subroutine is called, the stack pointer is pointing to the return address. Two bytes below it is the second parameter and four bytes below is the first parameter. The stack pointer is a reference point to these parameters. If the value of SP is captured in BP, then the return address is located at [bp+0], the second parameter is at [bp+2], and the first parameter is at [bp+4]. This is because SP and BP both had the same value and they both defaulted to the same segment, the stack segment.

This copying of SP into BP is like taking a snapshot or like freezing the stack at that moment. Even if more pushes are made on the stack decrementing the stack pointer, our reference point will not change. The parameters will still be accessible at the same offsets from the base pointer. If however the stack pointer increments beyond the base pointer, the references will become invalid. The base pointer will act as the datum point to access our parameters. However we have destroyed the original value of BP in the process, and this will cause problems in nested calls where both the outer and the ineer subroutines need to access their own parameters. The outer subroutine will have its base pointer destroyed after the call and will be unable to access its parameters. 

To solve both of these problems, we reach at the standard way of accessing parameters on the stack. The first two instructions of any subroutines accessing its parameters from the stack are given below.

push bp
mov  bp, sp

As a result our datum point has shifted by a word. Now the old value of BP will be contained in [bp] and the return address will be at [bp+2]. The second parameters will be [bp+4] while the first one will be at [bp+6]. We give an example of bubble sort subroutine using this standard way of argument passing through stack.

	; bubble sort subroutine taking parameters from stack

[org 0x0100]                     

              jmp start

data:         dw   60, 55, 45, 50, 40, 35, 25, 30, 10, 0

data2:        dw   328, 329, 898, 8923, 8293, 2345, 10, 877, 355, 98

              dw   888, 533, 2000, 1020, 30, 200, 761, 167, 90, 5

swapflag:     db   0

bubblesort:   push bp                 ; save old value of bp

              mov  bp, sp             ; make bp our reference point

              push ax                 ; save old value of ax

              push bx                 ; save old value of bx

              push cx                 ; save old value of cx

              push si                 ; save old value of si

              mov  bx, [bp+6]         ; load start of array in bx

              mov  cx, [bp+4]         ; load count of elements in cx

              dec  cx                 ; last element not compared

              shl  cx, 1              ; turn into byte count

mainloop:     mov  si, 0              ; initialize array index to zero

              mov  byte [swapflag], 0 ; reset swap flag to no swaps

innerloop:    mov  ax, [bx+si]        ; load number in ax

              cmp  ax, [bx+si+2]      ; compare with next number

              jbe  noswap             ; no swap if already in order

              xchg ax, [bx+si+2]      ; exchange ax with second number

              mov  [bx+si], ax        ; store second number in first

              mov  byte [swapflag], 1 ; flag that a swap has been done

noswap:       add  si, 2              ; advance si to next index

              cmp  si, cx             ; are we at last index

              jne  innerloop          ; if not compare next two

              cmp  byte [swapflag], 1 ; check if a swap has been done   

              je   mainloop           ; if yes make another pass 

              pop  si                 ; restore old value of si

              pop  cx                 ; restore old value of cx

              pop  bx                 ; restore old value of bx

              pop  ax                 ; restore old value of ax

              pop  bp                 ; restore old value of bp

              ret  4                  ; go back and remove two params

start:        mov  ax, data 

              push ax                 ; place start of array on stack

              mov  ax, 10 

              push ax                 ; place element count on stack

              call bubblesort         ; call our subroutine

              mov  ax, data2 

              push ax                 ; place start of array on stack

              mov  ax, 20

              push ax                 ; place element count on stack

              call bubblesort         ; call our subroutine again

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	The value of the stack pointer is captured in the base pointer. With further pushes SP will change but BP will not and therefore we will read parameters from bp+4 and bp+6.

The form of RET that takes an argument is used causing four to be added to SP after the return address has been popped in the instruction pointer. This will effectively discard the parameters that are still there on the stack. 

We push the address of the array we want to sort followed by the count of elements. As immediates cannot be directly pushed in the 8088 architecture, we first load it in the AX register and then push the AX register on the stack.


Inside the debugger, concentrate on the operation of BP and the stack. The parameters are placed on the stack by the caller, the subroutine accesses them using the base pointer, and the special form of RET removes them without any extra instruction. The value of stack pointer of FFF6 is turned into FFFE by the RET instruction. This was the value in SP before any of the parameters was pushed.

Stack Clearing by Caller or Callee

Parameters pushed for a subroutine are a waste after the subroutine has returned. They have to be cleared from the stack. Either of the caller and the callee can take the responsibility of clearing them from there. If the callee has to clear the stack it cannot do this easily unless RETn exists. That is why most general processors have this instruction. Stack clearing by the caller needs an extra instruction on behalf of the caller after every call made to the subroutine, unnecessarily increasing instructions in the program. If there are thousand calls to a subroutine the code to clear the stack is repeated a thousand times. Therefore the prevalent convention in most high level languages is stack clearing by the callee; even though the other convention is still used in some languages. 

If RETn is not available, stack clearing by the callee is a complicated process. It will have to save the return address in a register, then remove the parameters, and then place back the return address so that RET will function. When this instruction was introduced in processors, only then high level language designers switched to stack clearing by the callee. This is also exactly why RETn adds n to SP after performing the operation of RET. The other way around would be totally useless for our purpose. Consider the stack condition at the time of RET and this will become clear why this will be useless. Also observe that RETn has discarded the arguments rather than popping them as they were no longer of any use either of the caller or the callee.

The strong argument in favour of callee cleared stacks is that the arguments were placed on the stack for the subroutine, the caller did not needed them for itself, so the subroutine is responsible for removing them. Removing the arguments is important as if the stack is not cleared or is partially cleared the stack will eventually become full, SP will reach 0, and thereafter wraparound producing unexpected results. This is called stack overflow. Therefore clearing anything placed on the stack is very important. 

1.1. Local Variables

Another important role of the stack is in the creation of local variables that are only needed while the subroutine is in execution and not afterwards. They should not take permanent space like global variables. Local variables should be created when the subroutine is called and discarded afterwards. So that the spaced used by them can be reused for the local variables of another subroutine. They only have meaning inside the subroutine and no meaning outside it. 

The most convenient place to store these variables is the stack. We need some special manipulation of the stack for this task. We need to produce a gap in the stack for our variables. This is explained with the help of the swapflag in the bubble sort example.

The swapflag we have declared as a word occupying space permanently is only needed by the bubble sort subroutine and should be a local variable. Actually the variable was introduced with the intent of making it a local variable at this time. The stack pointer will be decremented by an extra two bytes thereby producing a gap in which a word can reside. This gap will be used for our temporary, local, or automatic variable; however we name it. We can decrement it as much as we want producing the desired space, however the decrement must be by an even number, as the unit of stack operation is a word. In our case we needed just one word. Also the most convenient position for this gap is immediately after saving the value of SP in BP. So that the same base pointer can be used to access the local variables as well; this time using negative offsets. The standard way to start a subroutine which needs to access parameters and has local variables is as under.

push bp
mov  bp, sp
sub  sp, 2

The gap could have been created with a dummy push, but the subtraction makes it clear that the value pushed is not important and the gap will be used for our local variable. Also gap of any size can be created in a single instruction with subtraction. The parameters can still be accessed at bp+4 and bp+6 and the swapflag can be accessed at bp-2. The subtraction in SP was after taking the snapshot; therefore BP is above the parameters but below the loacal variables. The parameters are therefore accessed using positive offsets from BP and the local variables are accessed using negative offsets. 

We modify the bubble sort subroutine to use a local variable to store the swap flag. The swap flag remembered whether a swap has been done in a particular iteration of bubble sort.

	
	Example 5.6

	01

02

03

04

05

06

07

08

09

10

11

12

13

14

15

16

17

18

19

20

21

22

23

24

25

26

27

28

29

30

31

32

33

34

35

36

37

38

39

40

41

42

43

44

45

46

47

48

49

50

51

52

53

54

55

56

57

58

59

60

61
	; bubble sort subroutine using a local variable

[org 0x0100]                     

              jmp start

data:         dw   60, 55, 45, 50, 40, 35, 25, 30, 10, 0

data2:        dw   328, 329, 898, 8923, 8293, 2345, 10, 877, 355, 98

              dw   888, 533, 2000, 1020, 30, 200, 761, 167, 90, 5

bubblesort:   push bp                 ; save old value of bp

              mov  bp, sp             ; make bp our reference point

              sub sp, 2               ; make two byte space on stack

              push ax                 ; save old value of ax

              push bx                 ; save old value of bx

              push cx                 ; save old value of cx

              push si                 ; save old value of si

              mov  bx, [bp+6]         ; load start of array in bx

              mov  cx, [bp+4]         ; load count of elements in cx

              dec  cx                 ; last element not compared

              shl  cx, 1              ; turn into byte count

mainloop:     mov  si, 0              ; initialize array index to zero

              mov  word [bp-2], 0     ; reset swap flag to no swaps

innerloop:    mov  ax, [bx+si]        ; load number in ax

              cmp  ax, [bx+si+2]      ; compare with next number

              jbe  noswap             ; no swap if already in order

              xchg ax, [bx+si+2]      ; exchange ax with second number

              mov  [bx+si], ax        ; store second number in first

              mov  word [bp-2], 1     ; flag that a swap has been done

noswap:       add  si, 2              ; advance si to next index

              cmp  si, cx             ; are we at last index

              jne  innerloop          ; if not compare next two

              cmp  word [bp-2], 1     ; check if a swap has been done   

              je   mainloop           ; if yes make another pass 

              pop  si                 ; restore old value of si

              pop  cx                 ; restore old value of cx

              pop  bx                 ; restore old value of bx

              pop  ax                 ; restore old value of ax

              mov  sp, bp             ; remove space created on stack

              pop  bp                 ; restore old value of bp

              ret  4                  ; go back and remove two params

start:        mov  ax, data 

              push ax                 ; place start of array on stack

              mov  ax, 10 

              push ax                 ; place element count on stack

              call bubblesort         ; call our subroutine

              mov  ax, data2 

              push ax                 ; place start of array on stack

              mov  ax, 20

              push ax                 ; place element count on stack

              call bubblesort         ; call our subroutine again

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	A word gap has been created for swap flag. This is equivalent to a dummy push. The registers are pushed above this gap.

The swapflag is accessed with [bp-2]. The parameters are accessed in the same manner as the last examples.

We are removing the hole that we created. The hole is removed by restoring the value of SP that it had at the time of snapshot or at the value it had before the local variable was created. This can be replaced with “add sp, 2” however the one used in the code is prefereed since it does not require to remember how much space for local variables was allocated in the start. After this operation SP points to the old value of BP from where we can proceed as usual.


We needed memory to store the swap flag. The fact that it is in the stack segment or the data segment doesn’t bother us. This will just change the addressing scheme. 

Exercises

1. Replace the following valid instruction with a single instruction that has the same effect. Don’t consider the effect on flags.

    push word L1
    jmp L2
L1:

2. Replace the following invalid instructions with a single instruction that has the same effect.

a. pop  ip

b. mov  ip, L5

c. sub  sp, 2
mov  [ss:sp], ax

d. mov ax, [ss:sp]
add sp, 2

e. add  sp, 6
mov  ip, [ss:sp-6]

3. Write a recursive function to calculate the fibonaccii of a number. The number is passed as a parameter via the stack and the calculated fibonaccii number is returned in the AX register. A local variable should be used to store the return value from the first recursive call. Fibonaccii function is defined as follows:

fibonaccii(0) = 0
fibonaccii(1) = 1
fibonaccii(n) = fibonaccii(n-1) + fibonaccii(n-2)

4. Write the above fibonaccii function iteratively. 

HINT: Use two registers to hold the current and the previous fibonaccii numbers in a loop.

5. Write a function switch_stack meant to change the current stack and will be called as below. The function should destroy no registers.

push word [new_stack_segment]
push word [new_stack_offset]
call switch_stack

6. Write a function “addtoset” that takes offset of a function and remembers this offset in an array that can hold a maximum of 8 offsets. It does nothing if there are already eight offsets in the set. Write another function “callset” that makes a call to all functions in the set one by one.

7. Do the above exercise such that “callset” does not use a CALL or a JMP to invoke the functions. 

HINT: Setup the stack appropriately such that the RET will execute the first function, its RET execute the next and so on till the last RET returns to the caller of “callset.”

8. Make an array of 0x80 bytes and treat it as one of 0x400 bits. Write a function myalloc that takes one argument, the number of bits. It finds that many consecutive zero bits in the array, makes them one, and returns in AX the index of the first bit. Write another function myfree that takes two arguments, index of a bit in the array, and the number of bits. It makes that many consecutive bits zero, whatever their previous values are, starting from the index in the first argument.

9. [Circular Queue] Write functions to implement circular queues. Declare 16x32 words of data for 16 queues numbered from 0 to 15. Each queue has a front index, a rear index and 30 locations for data totaling to 32 words. Declare another word variable whose 16 bits correspond to the 16 queues and a 1 bit signals that the correspdoning queue is used and a 0 bit signals that it is free. Write a function “qcreate” that returns a queue number after finding a free queue or -1 if it failed. Write a function “qdestroy” that marks the queue as free. Write two other functions “qadd” and “qremove” that can add and remove items from the circular queue. The two functions return 0 if they failed and 1 otherwise.
10. [Linked List] Declare 1024 nodes of four bytes each. The first 2 bytes will be used for data and the next 2 bytes for storing the offset of another node. Also declare a word variable “firstfree” to store the offset of the first free node. Write the following five functions:

a. “init” chains all 1024 nodes into a list with offset of first node in firstfree, offset of the second node in the later two bytes of the first node and so on. The later two bytes of the last node contains zero. 

b. “createlist” returns the offset of the node stored in firstfree through AX. It sets firstfree to the offset stored in the later two bytes of that node, and it sets the later two bytes of that node to zero.

c. “insertafter” takes two parameters, the offset of a node and a word data. It removes one node from freelist just like “createlist” and inserts it after the said node and updates the new node’s data part.
d. “deleteafter” takes a node as its parameter and removes the node immediately after it in the linked list if there is one.
e. “deletelist” takes a node as its parameters and traverses the linked list starting at this node and removes all nodes from it and add them back to the free list.
